



The University of
Nottingham

UNITED KINGDOM • CHINA • MALAYSIA

The Undecidability of the Halting Problem

A Formal Proof in Agda

Submitted April 2016, in partial fulfilment of
the conditions of the award of the degree:
BSc (Hons) Computer Science.

Zongzhe Yuan: psyzy3

4234822

With Supervision from Thorsten Altenkirch

School of Computer Science and Information Technology
University of Nottingham

**I hereby declare that this dissertation is all my own work, except as
indicated in the text:**

Signature _____

Date ____/____/____

Acknowledgements

Some People helped with the preparation of the paper. Three in particular have made significant contributions to the content of the paper, the style of the paper, as well as the code of the proof in *Agda*: Gabe Dijkstra helped a lot with the professional *Agda* technique. Ambrus Kaposi helped significantly with many proofs and several lemmas involving some key points in the proof of the correctness of the universal model and the proof of the *Halting Problem*. Thorsten Altenkirch, as the supervisor of the project, provided the overview and the general idea of the proof of the *Halting Problem* in *Agda*, and helped me a lot during the whole project. Part of the definition, including the definition of *WHILE* language that was used in the project follows the definition in the paper *Computability and Complexity: from a Programming Perspective* by Neil D. Jones in 1997[?].

The complete code could be found in my *GitHub* or go to the next url: <https://github.com/ZongzheYuan/HaltingProblem.git>.

Abstract

The *Halting Problem*, is a famous problem in the computability theory. Alan Turing proved the undecidability of the *Halting Problem* in 1936[?]. By choosing some computational models which are *Turing-equivalents* to the *Turing Machine*[?], defining the syntax and semantics of that model and constructing the universal computational model based on the theory of *Turing-complete*, one could prove the undecidability of the *Halting Problem*. The aim of this project is constructing a proper computational model, proving the correctness of the universal model (has the property of *Turing completeness*), and finally using that model to prove the undecidability of the *Halting Problem*. In more details, the project will develop a machine checked proof of the undecidability of the *Halting Problem* using the *Agda* system.

Keywords: Agda, Turing-completeness, Universal Model, Undecidable, Machine checked proof, WHILE language

Contents

1	Approach to the Project	1
1.1	Overview	1
1.2	Background	2
1.2.1	The Church-Turing Thesis	2
1.2.2	Decidable and Non-Decidable	4
1.2.3	The Halting Problem	4
1.2.4	The WHILE language	5
1.2.5	Agda	5
1.3	Related Work	6
1.3.1	Prove the Undecidability of the Halting Problem	6
2	Design and Implementation	8
2.1	The WHILE Language in <i>Agda</i>	8
2.1.1	Tree Data Structure	8
2.1.2	Syntax	9
2.1.2.1	Expression	9
2.1.2.2	Command	9
2.1.2.3	Program	10
2.1.3	Semantics	10
2.1.3.1	Environment	10
2.1.3.2	The Semantics of Expression	11
2.1.3.3	The Semantics of Command	11
2.1.3.4	The Semantics of Program	12
2.1.4	Run WHILE in K Steps	12
2.2	The Universal WHILE model	13
2.2.1	Interpretation of the WHILE program	13
2.2.1.1	Code the Expression	14
2.2.1.2	Code the Command	14

2.2.1.3	Code the Program	14
2.2.1.4	Decode	15
2.2.2	The Universal WHILE model	15
2.2.2.1	Interpretation in Agda	16
2.2.2.2	Interpretation in the universal WHILE program . . .	18
2.2.3	Correctness of the Universal WHILE model	19
2.3	Proof to the Halting Problem	19
2.3.1	Construct the WHILE program U	19
2.3.1.1	Property 1 of U	20
2.3.1.2	Property 2 of U	21
2.3.2	Proof the Undecidability of the Halting Problem	21
2.3.2.1	Property 1 of Machine H	21
2.3.2.2	Property 2 of Machine H	21
2.3.2.3	Propositional Proof	22
2.3.2.4	Final Proof	22
3	Evaluation, Summary and Further Work	24
3.1	Evaluation and Summary	24
3.2	Main Difficulties	26
3.3	Further Work	27
3.3.1	WHILE Program Variables Transformation	27
3.3.2	Interpretation of the WHILE Program with Arbitrary Variables	29
A	Definition of the WHILE Language	30
A.1	Syntax of the Expression	30
A.2	Syntax of the Command	30
A.3	Semantics of the Expression	30
A.4	Semantics of the Command	31
A.5	Example of the WHILE program	31
A.5.1	Example of the WHILE program in the WHILE model	31
A.5.2	Example of the WHILE program in <i>Agda</i>	32
A.5.3	Execution of the Example <i>WHILE</i> program in <i>Agda</i>	32

B	Definition of the Universal WHILE model	35
B.1	Constant in WHILE	35
B.2	Code the WHILE program	36
B.3	Variable of the Universal WHILE Program	37
B.4	Universal WHILE Program	38
B.5	STEP Macro	38
B.6	Constant in the Expression	39

Chapter 1

Approach to the Project

1.1 Overview

Alan Turing analysed and formalised the class of all computational procedure in 1936[?] when the well-known computational model *Turing Machine* was introduced to the world. *Alan Turing* has used the model of the *Turing Machine* to answer the question "Does a machine exist that can determine whether any arbitrary machine on its tape is 'circular'", and proved the undecidability of the *Halting problem* in 1936 as well[?], which constitute the objective of this paper.

To prove the undecidability of the *Halting problem*, one must use a computational model that is *Turing-equivalent* to the *Turing Machine* in the thesis of the *Church-Turing thesis*[?]. To simplify, we choose a computational model that contains one *input*, one *output* and a *function* from *input* to *output*. The detail definition will be shown in 2.1. We choose some proper data structure D , defined in 2.1.1, to represent the data and let the *input* and *output* belongs to D . Then we could define the *function* (in 2.1.3) as a relation $\bullet \rightarrow \bullet \subseteq D \times D$ which is also the mapping from the *input* to the *output*. To prove the correctness of the relation, we must construct this computational model with concrete syntax and semantics by using some proof assistant language – the meta language (in 2.1.2 and 2.1.3).

Then we should construct the universal model following the thesis of *Turing-complete*[?] (in 2.2). That is, if we have $f \in \text{function}$, $\text{input} \in D$, and $f(\text{input}) \equiv \text{output}$, then the universal model u has the property that $u(\lfloor f \rfloor \cdot \text{input}) \equiv \text{output}$ where $\lfloor f \rfloor$ is the code of *function* f in data format D and \cdot is the concat symbol in the data structure D .

In order to prove the correctness of our predefined universal computational model,

we should define the concrete code method to code the *function* into our predefined data structure D without ambiguity (in 2.2.1). The first step in our process is to construct the interpretation simulation step s in the chosen meta language and prove the relation between the *function* f and the interpretation simulation s (in 2.2.2). We could define the interpretation simulation as a stack machine: $(Command, Stack, Variable) \Rightarrow (Command', Stack', Variable') \subseteq (D, D, D) \times (D, D, D)$ which is a one-step relation between two triples. Then we should prove that if we have $f \in function$, $input$ and $output \in D$, and $f(input) \equiv output$, then we can get $(\lfloor f \rfloor, \epsilon, input) \Rightarrow^* (\epsilon, \epsilon, output)$ where \Rightarrow^* is multistep relationship of relation \Rightarrow (in 2.2.2.1). From the proof given above we know that we can use the meta language to simulate the one step of the *function* in several steps. Then we can construct the universal model, and prove that the interpretation step simulated in the meta language has correspondence to the interpretation in the universal model (in 2.2.2.2). If we have $f \in function$, $input$ and $output \in D$, and $(\lfloor f \rfloor, \epsilon, input) \Rightarrow^* (\epsilon, \epsilon, output)$, then the universal model u has the property that $u(\lfloor f \rfloor \cdot input) \equiv output$. The proof is inducted by the step of relation \Rightarrow (in 2.2.3). Finally we can conclude that the program and the universal model has the correspondence that for $f \in function$, $input$ and $output \in D$, and $f(input) \equiv output$, then for the universal model u , $u(\lfloor f \rfloor \cdot input) \equiv output$.

To prove the undecidability of the *Halting Problem*, we can assume that there exists a program $h \in \text{Program}$ that has some properties: $\forall p \in \text{Program}$ and $\forall input \in \text{Data}$, if p halt on $input$, then $h(\lfloor p \rfloor \cdot input) \equiv 1$, else $h(\lfloor p \rfloor \cdot input) \equiv 0$, which means h decides the *Halting Problem*. Then we can construct a program $m \in \text{Program}$ and feed h to m which means running h inside m . Then we let h to decide whether $m(\lfloor m \rfloor)$ will halt or not, which will end up in contradiction with the definition of m . The construction and the proof could be found in 2.3. Finally we can conclude that h doesn't exist by contradiction to our premise, and there is no model that can decide the *Halting Problem*, which otherwise means that the *Halting Problem* is undecidable.

1.2 Background

1.2.1 The Church-Turing Thesis

The *Church-Turing thesis* is a hypothesis about the nature of computable functions[?]. The thesis states that every effective computation can be carried out by a *Turing Machine*[?]. Turing gave the definition of his thesis as the LCMs [logical computing

machines: Turing's expression for *Turing Machines*] can do anything, that could be described as "rule of thumb" or "purely mechanical" (Turing 1948:7.)[?].

The two basic concepts that related to this paper are the *Turing completeness* and the *Turing equivalence*.

Turing completeness is the concept in the computability theory, such a computational model (for example a programming language, or recursive function) may satisfy *Turing completeness* if and only if the model could be used to simulate any single-taped *Turing Machine*[?].

Turing equivalence said that if two computational models could simulate each one by the other, then these two computational machine is called *Turing equivalence*[?].

By the *Church-Turing Thesis*, any function that can be computed by some algorithm can be computed by a *Turing Machine*[?]. Thus there are many computational models that is *Turing equivalent* to a *Turing Machine*[?]. Though the goal of the project is to prove the undecidability of the *Halting Problem*, constructing and formalising a universal *Turing Machine* is quite complex. Thus we can choose many other notions of effective other than the *Turing Machine* (means that the notion is *Turing-complete*):

- Recursive functions as defined by Kleene[?]
- The lambda calculus approach to function definitions due to Church[?]
- Random access machines[?]
- Markov algorithms[?]

These different forms that contain same efficient computing modules share several common characteristics[?].

- The procedure consists of finite size of instructions.
- The computation is carried out in a discrete stepwise fashion but not in the method of continuous methods with analogue devices.
- The computation is carried out deterministically but not in random methods.

- Though a terminating computation must not rely on infinite amount of space or time, there is no limitation set for the amount of memory storage space or time available.

Those computational model listed above have been proved to be *Turing-equivalence* to the *Turing Machine*, which means they also have the property of *Turing completeness* and can construct universal computational model.

Universal Turing Machine is a Turing Machine that can simulate an arbitrary *Turing Machine* on arbitrary input[?]. That is for $t \in \mathbf{TM}$, $input$ and $output \in \Sigma^*$ (the set of input symbols on Kleene star[?]) , and $t(input)$ yields $output$, then the universal *Turing Machine* $u \in \mathbf{UTM}$ has the property that $u([t] \cdot input)$ yields $output$ where $[t]$ is the code of *Turing Machine* t .

1.2.2 Decidable and Non-Decidable

In the area of computability, a set \mathbf{S} is **Recursive (Decidable)** \iff given a set \mathbf{D} and $\mathbf{S} \subseteq \mathbf{D}$, there is a function f applies to the element $a \in \mathbf{D}$, f will return "true" if $a \in \mathbf{S}$ and f will return "false" if $a \notin \mathbf{S}$ [?]. **Decidable** set is closed under union, intersection, complement difference and Kleene star[?] circumstances.

A set \mathbf{S} is **Recursively Enumerable (Semi-Decidable)** \iff given a set \mathbf{D} and $\mathbf{S} \subseteq \mathbf{D}$, there is a function f applies to the element $a \in \mathbf{D}$, f will return "true" if $a \in \mathbf{S}$ and f will return "false" or never terminate if $a \notin \mathbf{S}$ which means no guarantee to terminate under the element $a \notin \mathbf{S}$ [?]. If a set \mathbf{S} is recursively enumerable and the implement of \mathbf{S} is also recursively enumerable, then set \mathbf{S} is Recursive[?].

1.2.3 The Halting Problem

In the area of computability theory, the *Halting Problem* indicates that a given universal computing program (the model that is *Turing-complete*[?]) may determine any other arbitrary computing program that would return the result on arbitrary input in a finite number of steps (a finite period of time), or would run forever[?]. That is, $\forall t \in \mathbf{TM}$, $input \in \Sigma^*$, if there exists $h \in \mathbf{TM}$ such that, if $halt_t(input)$, then $h([t] \cdot input) \equiv true$, else $h([t] \cdot input) \equiv false$, then h determining the *Halting Problem*.

It is easily to prove that the *Halting Problem* is semi-decidable[?] because $\forall t \in \mathbf{TM}$, $input \in \Sigma^*$, if eventually $halt_t(input)$ then we can easily get the result. However

whether the *Halting Problem* is decidable or not is interesting, and the aim of this paper is to prove the undecidability of the *Halting Problem*.

1.2.4 The WHILE language

The *WHILE* language is a language that was just the right mixture of expressive power and simplicity. It provides the strict definitions of syntax and semantics and stays in the same level with *Turing Machine* model in terms of computing effectiveness (Turing-complete)[?]. As well the data structure of *WHILE* treats the program as data object which could solve some rather complex missions. Furthermore, with the simplicity, *WHILE* language could be simply used to prove many theorems and their behaviours. By considering those several reasons, the project will focus on proving the *Halting Problem* undecidability in the model of *WHILE* language[?]. The definition of the *WHILE* language can be found in appendix A. The implementation of the *WHILE* language in *Agda* can be found in 2.1.

1.2.5 Agda

Agda, a dependent type language and an interactive proof assistant that implements the Martin-Löf type theory[?], could assist me to develop a machine checked proof and formalise the proof of the *Halting Problem*. Because the *dependent type* allow types to talk about values, the programs written by those languages could be encoded properties of values as types whose elements are proofs that the property is true, which means that a dependently typed programming language can be used as a logic, and is needed to be integrated, not crash or non-terminate. And the mathematical proofs in *Agda* are written as structural induction format, which are recursive functions that inducing on some inductive type argument. Thus, it is equivalent to give mathematical proof by constructing some well-typed function that could finally terminate. Therefore, *Agda* can be used as a framework to formalise formal logic systems and to prove the lemma which can be proved in mathematical ways.

We can use the key word **data** in *Agda* to define some inductive types, or more generally define some inductive families. Here we define the basic type *natural number* as an example.

```
data ℕ : Set where
  zero : ℕ
  suc   : ℕ → ℕ
```

Then we can define the function *plus* based on the *natural number*.

```

_+_ : ℕ → ℕ → ℕ
_+_ zero b = b
_+_ (suc a) b = suc (a + b)

```

Finally we can prove the *associative law* of our defined function by doing induction on the first argument, which is an inductive type in our definition.

```

suc-ok : {a b : ℕ} → suc a + b ≡ suc (a + b)
suc-ok {zero} = refl
suc-ok {suc a} = cong suc refl

plus-asso : {a b c : ℕ} → (a + b) + c ≡ a + (b + c)
plus-asso {zero} = refl
plus-asso {suc a}{b}{c} = (suc a + b) + c
                        =⟨ cong (λ x → x + c) (suc-ok {a}{b}) ⟩
                        suc (a + b) + c
                        =⟨ suc-ok {a + b}{c} ⟩
                        (suc ((a + b) + c))
                        =⟨ cong suc (plus-asso {a}{b}{c}) ⟩
                        suc (a + (b + c))
                        ■

```

Using *Agda* as the proof assistant language to this project has a lot of advantages. Another convenient virtue is that we can use Unicode characters while we write program in *Agda*, which let the proofs in *Agda* look similar to those logic proofs on paper and on textbooks.

The concept and those basic technique of *Agda* could be found in the book *Dependently Typed Programming in Agda* by *Ulf Norell and James Chapman* in 2009[?].

1.3 Related Work

1.3.1 Prove the Undecidability of the Halting Problem

In 1936, *Alan Turing* has proved that a general algorithm to solve the *Halting Problem* for all possible program-input pairs cannot exist[?].

To prove the undecidability of the *Halting Problem*, we firstly assume that there exists some computational models that could decide the *Halting Problem*, then we can found some contradiction which prove that there is no computational model that could decide the *Halting Problem*, which means the *Halting Problem* is undecidable.

We could represent the decision problems as the set of objects that have the property in question.

Then the *Halting Problem* could be represent as the halting set $H = \{(p, inp) \mid \text{program } p \text{ halts when it runs on input } inp\}$. Then we could prove that the following function h is not computable which means there is no total computable function that may decide whether an arbitrary program p halts on arbitrary input inp or not[?]:

Suppose there exists a *Turing Machine* which could decide the *Halting Problem*:

$$h(i, x) = \begin{cases} 1, & \text{if program } i \text{ halts on } x \\ 0, & \text{otherwise} \end{cases}. \text{ Then we construct a universal } Turing \text{ Ma-}$$

chine $u(x)$ which could take the binary code of another *Turing Machine* as input such that inside u we run h on (x, x) and if the result of h is 1 then u will loop forever, otherwise u will halt. Lastly, we run u on $\lfloor u \rfloor$ that is the binary code of the *Turing Machine* u , which means inside u *Turing Machine* h will run on $(\lfloor u \rfloor, \lfloor u \rfloor)$. Nevertheless, if the universal *Turing Machine* u finally halt on its binary code, then the *Turing machine* h will return 1, but u will loop forever by the definition of *Turing Machine* u . Thus, u can't halt on its binary code. However in that situation h will return 0 which means u will halt finally. In conclusion, there is no universal *Turing Machine* that could decide the *Halting Problem* which can be proved by finding the contradiction to the assumption[?].

Chapter 2

Design and Implementation

2.1 The WHILE Language in *Agda*

The definition of the *WHILE* language mentioned in this section follows the definition in the paper *Computability and Complexity: from a Programming Perspective* by Neil D. Jones in 1997[?].

2.1.1 Tree Data Structure

The language *WHILE* that computes with a *trees* data structure is built from a finite set. Thus we define a tree data structure \mathbb{D} with several related functions in *Agda* at first. To achieve it, we must define the *atoms* for the *trees* beforehand. *Atoms* mean small substances that can't be divided further into subparts. However, the complexity in defining a large number of *atoms* may make our proof become more complicated. In fact, we can define only one *atom* called *nil*, because any other values, or say other "atoms" that we presume to define could be constructed by combining different number of *nil* in different order. Thus, we define the data structure as:

```
data D : Set where
  dnil : D
  _•_   : D → D → D
```

And provide the approach to visit the first or the second element of an element in \mathbb{D} .

```
dfst : D → D
dfst dnil = dnil
dfst (d1 • d2) = d1
```

```

dsnd : D → D
dsnd dnil = dnil
dsnd (d1 • d2) = d2

```

2.1.2 Syntax

The syntax of *WHILE* is defined following the syntax definition in the book[?], and is consisted of *expression*, *command* and *program*.

2.1.2.1 Expression

The *expression* is constructed in a binary tree format, which has the same format of the data structure we defined previously. Then we can define the *expression* as the value of some variable, the *atom* value which is the *nil*, the first or the second value of another *expression*, the combination of two *expressions*, or the equality of two *expressions*. The definition of the syntax of *expression* could be found in A.1.

Then we define the data type of \mathbb{E} in *Agda* :

```

data E (n : ℕ) : Set where
  var   : Fin n → E n
  nil   : E n
  cons  : E n → E n → E n
  hd    : E n → E n
  tl    : E n → E n
  _=?_  : E n → E n → E n

```

We use the member in a **finite set** to represent the variables instead of variable names. For example **Fin n** is a finite set that contains **n** elements from **zero** to **suc (suc ... suc (zero))**. Then we can directly use the element in the set to indicate variables' names (**zero** is the first variable and so on).

2.1.2.2 Command

A *command* is either the assignment from some *expressions* to some variables, or the sequence of two *commands*, or the *while* loop. The definition of the syntax of *command* could be found in A.2.

Then we define the data type of \mathbb{C} in *Agda* :


```

data C (n : ℕ) : Set where
  _:=_      : Fin n → E n → C n
  _→→_     : C n → C n → C n
  while    : E n → C n → C n

```

2.1.2.3 Program

The *program* is consist of an *input* variable which is the variable to store the *input*, an *output* variable which is the variable to store the final result, and a *command*:

Programs $\ni P ::= \text{read } X ; C ; \text{write } Y$

And we can define the same data type of \mathbb{P} in *Agda* :

```

data P (n : ℕ) : Set where
  prog : Fin n → C n → Fin n → P n

```

2.1.3 Semantics

To define the semantics of *WHILE* language, we must give a definition of the *Partial Function* at first[?]:

Let A, B be sets, a partial function g is written as $g : A \rightarrow B_{\perp}$ and we said g is effectively computable if there is an effective procedure such that for any $x \in A$:

- The procedure eventually halts, yielding $g(x) \in B$, if $g(x)$ is defined;
- The procedure never halts, if $g(x)$ is undefined.

Then we could show that the program in *WHILE* can be used as a partial function from \mathbb{D} to \mathbb{D}_{\perp} .

2.1.3.1 Environment

We should define the *environment* of the *command*, written as $[x_1 \mapsto v_1, x_2 \mapsto v_2, \dots, x_n \mapsto v_n]$ to indicate the finite mapping function such that $s(x_i) = v_i$, where $v_i \in \mathbb{D}$. Then we use the notion σ to indicate the *environment* in *WHILE* that has type *Store*, and for $p \in \mathbb{P}$, $p = \text{read } X ; C ; \text{write } Y$, the initial store σ_0^p is $[X \mapsto d, Y_1 \mapsto \text{nil}, \dots, Y_n \mapsto \text{nil}]$, and \forall variable X and Z such that X and Z are variables in program p and $X \neq Z$, then Z is in Y_i .

In *Agda* , we use the data type vector **Vec D n** to represent the store. Vector in *Agda* have the type:

```
data Vec {a} (A : Set a) : ℕ → Set
```

which bind a list of certain type element with certain number of length. Because the *store* and the program use the same \mathbf{n} for both finite set of variables and their correspondence values, the program is impossible to meet the condition that one variable hasn't been defined.

2.1.3.2 The Semantics of Expression

Then we can define the evaluation function ε with the type of $\mathbb{E} \longrightarrow (Store^P \rightarrow \mathbb{D})$, which means for $e \in \mathbb{E}$ and a given *store* of program P : $\sigma \in Store^P$, $\varepsilon[e]\sigma = d \in \mathbb{D}$. The definition of the evaluation function could be found in A.3. The *Agda* implementation of the evaluation function is defined as following:

```
eval : {n : ℕ} → E n → Vec D n → D
eval (var x) v = dlookup x v
eval nil v = dnil
eval (cons e1 e2) v = eval e1 v • eval e2 v
eval (hd e) v = dhead (eval e v)
eval (tl e) v = dtail (eval e v)
eval (e1 =? e2) v with equalD? (eval e1 v) (eval e2 v)
eval (e1 =? e2) v | eq x = dnil • dnil
eval (e1 =? e2) v | neq x = dnil
```

2.1.3.3 The Semantics of Command

The execution of a *command* in the program P could be used as a function $f : \mathbb{C} \times Store^P \times Store^P$. However, because we can't guarantee that the execution of a *command* will eventually halt and yield some output, the execution function should be a partial function $f : \mathbb{C} \times Store^P \times Store^P_{\perp}$. From this point of view, it is better to use a relation instead of a partial function to represent the execution of a *command* $c \in \mathbb{C}$ as $c \vdash \sigma \rightarrow \sigma' \subseteq \mathbb{C} \times Store^P \times Store^P$ where σ' is the new *environment* updated by the execution of *command* c . The definition of the execution *command* relationship could be found in A.4. The *Agda* implementation of the execution relationship is defined as following:

```
data _⊢⇒_ {n : ℕ} : C n → Vec D n → Vec D n → Set where
  whilef : {e : E n}{c : C n}{env : Vec D n}
    → isNil (eval e env)
    → (while e c) ⊢ env ⇒ env
  whilet : {e : E n}{c : C n}{env1 env2 env3 : Vec D n}
    → isTree (eval e env1)
```

```

→ c ⊢ env1 ⇒ env2
→ (while e c) ⊢ env2 ⇒ env3
→ (while e c) ⊢ env1 ⇒ env3
assign : {v : Fin n}{e : E n}{env : Vec D n}
→ (v := e) ⊢ env ⇒ (updateE v (eval e env) env)
seq      : {c1 c2 : C n}{env1 env2 env3 : Vec D n}
→ c1 ⊢ env1 ⇒ env2
→ c2 ⊢ env2 ⇒ env3
→ (c1 →→ c2) ⊢ env1 ⇒ env3

```

2.1.3.4 The Semantics of Program

Similar to the definition of execution of \mathbb{C} in *WHILE*, the execution of program \mathbb{P} should also be defined as a relationship. Following the definition given by *Neil D. Jones* in the paper[?], we could know that the semantics of *WHILE* is:

$\llbracket \bullet \rrbracket^{\text{WHILE}} : \mathbb{P} \rightarrow (\mathbb{D} \rightarrow \mathbb{D}_{\perp})$ defined for $p = \text{read } X; C; \text{write } Y$ by:

$\llbracket p \rrbracket^{\text{WHILE}} = e$ if $C \vdash \sigma_0^p(d) \rightarrow \sigma$ and $\sigma(Y) = e$

If there is no e such that $\llbracket p \rrbracket = e$, then p **loops** on d ; otherwise p terminates on d .

Following the definition, we define the partial relationship in *Agda* as following:

```

data ExecP {n : ℕ} : P n → D → D → Set where
  terminate : (x y : Fin n){c : C n}{env : Vec D n}{d : D}
    → c ⊢ (updateE x d initialVec) ⇒ env
    → ExecP (prog x c y) d (dlookup y env)

```

The example of *WHILE* program and the execution of *WHILE* program could be found at A.5.

2.1.4 Run WHILE in K Steps

Even the execution of the *command* \mathbb{C} and the *program* \mathbb{P} are both a partial relationship, which means that we can't guarantee the eventual halt of the *command* or the *program* on some input(that is what we are proving), we can still define the partial function that try to execute the *command* \mathbb{C} and *program* \mathbb{P} in k time.

Firstly we should define some data type to record the execution step number of a given command.

```

record ResultCT {n : ℕ}(c : C n)(inp : Vec D n) : Set where
  field
    out : Vec D n

```

```

exe  : c ⊢ inp ⇒ out
time : ℕ

```

Then we can construct the function to prove that one *command* may be executed in k time:

```

kStepC : {n : ℕ} → (k : ℕ) → (c : C n) → (inp : Vec D n)
        → (Maybe (ResultCT c inp))

```

The idea of that function is to do induction on the k at first. No *command* could be run in *zero* step. Then the function will do induction on the *command*. The assignment step will only take *1* step. The steps that need to take relies on the sequence of two *command* c_1 and c_2 will be the sum of steps that take on c_1 and the steps that costed on c_2 . Similarly in the *while* loop, the *command* will take *zero* step if the *expression* to the *while* loop could be evaluated to *false*. It will cost steps that costed on c and the continuous steps that costed on the following *while* loop as the total steps to the *command*.

Finally, if a command c could be executed in k steps, then a program $p = (\text{read } X; c; \text{write } Y)$ could also be executed in k steps:

```

kStepP : {n : ℕ} → (time : ℕ) → (p : P n) → (inp : D)
        → (Maybe (Σ D (ExecP p inp)))

```

2.2 The Universal WHILE model

2.2.1 Interpretation of the WHILE program

In order to construct the universal *WHILE* model in *Agda* later, we must define the method to code a program into \mathbb{D} in order to feed the program as the input to the universal *WHILE* program later. It is important to define the operator \bullet of our data structure \mathbb{D} with no association, which would avoid ambiguity.

Initially we should define some constants to indicate some distinct elements of \mathbb{D} . Those constants would represent the program in the form of \mathbb{D} . The definition could be found in B.1.

Then we defined the function that map the *program* to \mathbb{D} : $code \in \mathbb{P} \times \mathbb{D}$. The *code* function consists of three parts.

2.2.1.1 Code the Expression

Initially we should define the function that code the *expression* into \mathbb{D} . The definition of the mapping function of *expression* could be found in B.2. Then we could define the function in *Agda* following the same definition:

```
codeE : {n : ℕ} → E n → D
codeE (var x) = dvar • dftod x
codeE nil = dquote • dnil
codeE (cons e1 e2) = dcons • (codeE e1 • codeE e2)
codeE (hd e) = dhd • codeE e
codeE (tl e) = dtl • codeE e
codeE (e1 =? e2) = d=? • (codeE e1 • codeE e2)
```

2.2.1.2 Code the Command

Then we should define the function that code the *command* into \mathbb{D} . The definition of the mapping function of *command* could be found in B.2. Then we could define the function in *Agda* following the same definition:

```
codeC : {n : ℕ} → C n → D
codeC (x := e) = d:= • ((dvar • dftod x) • codeE e)
codeC (c1 →→ c2) = d→→ • (codeC c1 • codeC c2)
codeC (while e c) = dwhile • (codeE e • codeC c)
```

2.2.1.3 Code the Program

Finally we should define the function that code the *program* into \mathbb{D} . In the later version, I also add the number of variables of the program to the result of coding in order to allow the universal *WHILE* program could interpret any *WHILE* program with arbitrary variables in 3.3.1. The definition of the mapping function of *program* could be found in B.2. Then we could define the function in *Agda* following the same definition:

```
codeP : {n : ℕ} → P n → D
codeP {n} (prog x c y) = (dvar • dftod x) •
                        (codeC c •
                         (dvar • dftod y))
```

2.2.1.4 Decode

Beyond the coding method that maps the *program* to \mathbb{D} , I also define the function that decode the \mathbb{D} and map it to the *program*. However because the function *decode* is a partial function, sometimes it may cause decoding fail because the input \mathbb{D} doesn't follow the format of the program.

```
decodeE : {n : ℕ} → D → Maybe (E n)
decodeC : {n : ℕ} → D → Maybe (C n)
decodeP : D → Maybe (Σ ℕ P)
```

2.2.2 The Universal WHILE model

Initially we should define the variable in the *WHILE* language, the definition could be found in B.3.

Then we should define some syntax sugar such as *if* and *if-else*:

```
if8 : E 8 → C 8 → C 8
if8 e c = (Z := e) →→ while (var Z) ((Z := nil) →→ c)

if-else8 : E 8 → C 8 → C 8 → C 8
if-else8 e c1 c2 = (Z := e) →→
  (W := cons nil nil) →→
  ((while (var Z)
    ((Z := nil) →→
     (W := nil) →→
     c1))) →→
  (while (var W)
    ((W := nil) →→
     c2)))
```

Finally we should define the universal *WHILE* program. Here we firstly define the universal *WHILE* program that could simulate other *WHILE* program which has only one variable. The definition could be found in B.4.

Then the program is defined in *Agda* as:

```
universalI : P 8
universalI = prog PD ((Pp := hd (var PD))
  →→
  (Cc := hd (tl (var Pp)))
  →→
  (Cd := cons (var Cc) nil))
```

```

→→
(St := nil)
→→
(V1 := tl (var PD))
→→
(while (var Cd) STEP-I))
V1

```

2.2.2.1 Interpretation in Agda

We can imitate the simulation step following the definition using *Agda* at first. The definition of the *STEP* Macro could be found in B.5.

Initially we could define the data relationship $(Cd, St, V1) \Rightarrow (Cd', St', V1') \in (\mathbb{D}, \mathbb{D}, \mathbb{D}) \times (\mathbb{D}, \mathbb{D}, \mathbb{D})$ as a one step relationship.

```

data _⇒_ : D × D × D → D × D × D → Set where
  equote  : (d Cr St V1 : D)
    → < (dquote • d) • Cr , St , V1 >
    ⇒ < Cr , d • St , V1 >
  evar1   : (Cr St V1 : D)
    → < (dvar • dftod {1} zero) • Cr , St , V1 >
    ⇒ < Cr , V1 • St , V1 >
  ehd     : (E Cr St V1 : D)
    → < (dhd • E) • Cr , St , V1 >
    ⇒ < E • (dohd • Cr) , St , V1 >
  edohd   : (T Cr St V1 : D)
    → < dohd • Cr , T • St , V1 >
    ⇒ < Cr , (dfst T) • St , V1 >
  etl     : (E Cr St V1 : D)
    → < (dtl • E) • Cr , St , V1 >
    ⇒ < E • (dotl • Cr) , St , V1 >
  edotl   : (T Cr St V1 : D)
    → < dotl • Cr , T • St , V1 >
    ⇒ < Cr , (dsnd T) • St , V1 >
  econs   : (E1 E2 Cr St V1 : D)
    → < (dcons • (E1 • E2)) • Cr , St , V1 >
    ⇒ < E1 • (E2 • (docons • Cr)) , St , V1 >
  edocons : (U T Cr St V1 : D)
    → < docons • Cr , U • (T • St) , V1 >
    ⇒ < Cr , (T • U) • St , V1 >
  e=?     : (E1 E2 Cr St V1 : D)
    → < (d=? • (E1 • E2)) • Cr , St , V1 >
    ⇒ < E1 • (E2 • (do=? • Cr)) , St , V1 >
  edo=?   : (U T Cr St V1 : D)
    → < do=? • Cr , U • (T • St) , V1 >
    ⇒ < Cr , (dequal T U) • St , V1 >
  e→→    : (C1 C2 Cr St V1 : D)
    → < (d→→ • (C1 • C2)) • Cr , St , V1 >
    ⇒ < C1 • (C2 • Cr) , St , V1 >
  e:=     : (E Cr St V1 : D)

```

```

    → < (d := • ((dvar • dftod {1} zero) • E)) • Cr , St , V1 >
    ⇒ < E • (doasgn • Cr) , St , V1 >
edoasgn : (W Cr St V1 : D)
    → < doasgn • Cr , W • St , V1 >
    ⇒ < Cr , St , W >
ewhile : (E C Cr St V1 : D)
    → < (dwhile • (E • C)) • Cr , St , V1 >
    ⇒ < E • (dowh • ((dwhile • (E • C)) • Cr)) , St , V1 >
edowhf : (E C Cr St V1 : D)
    → < dowh • ((dwhile • (E • C)) • Cr) , dnil • St , V1 >
    ⇒ < Cr , St , V1 >
edowht : (E C X Y Cr St V1 : D)
    → < dowh • ((dwhile • (E • C)) • Cr) , (X • Y) • St , V1 >
    ⇒ < C • ((dwhile • (E • C)) • Cr) , St , V1 >
enil : (St V1 : D) → < dnil , St , V1 > ⇒ < dnil , St , V1 >

```

Then we should define the several steps relationship $(Cd, St, V1) \Rightarrow^* (Cd', St', V1') \in (\mathbb{D}, \mathbb{D}, \mathbb{D}) \times (\mathbb{D}, \mathbb{D}, \mathbb{D})$.

```

data _⇒*_ : D × D × D → D × D × D → Set where
  id : (Cr St V1 : D) → < Cr , St , V1 > ⇒* < Cr , St , V1 >
  seq : (Cr1 Cr2 Cr3 St1 St2 St3 V11 V12 V13 : D)
    → < Cr1 , St1 , V11 > ⇒ < Cr2 , St2 , V12 >
    → < Cr2 , St2 , V12 > ⇒* < Cr3 , St3 , V13 >
    → < Cr1 , St1 , V11 > ⇒* < Cr3 , St3 , V13 >

```

We should proof the associative of relation \Rightarrow^* .

```

⇒*-m : (Cr1 Cr2 Cr3 St1 St2 St3 V11 V12 V13 : D)
    → < Cr1 , St1 , V11 > ⇒* < Cr2 , St2 , V12 >
    → < Cr2 , St2 , V12 > ⇒* < Cr3 , St3 , V13 >
    → < Cr1 , St1 , V11 > ⇒* < Cr3 , St3 , V13 >
⇒*-b : (Cr1 Cr2 Cr3 St1 St2 St3 V11 V12 V13 : D)
    → < Cr1 , St1 , V11 > ⇒* < Cr2 , St2 , V12 >
    → < Cr2 , St2 , V12 > ⇒ < Cr3 , St3 , V13 >
    → < Cr1 , St1 , V11 > ⇒* < Cr3 , St3 , V13 >

```

Then we can prove that if for $E \in \mathbb{E}$, $\varepsilon[E][V1 \mapsto d] = d_e$, then $((codeEE \cdot Cr), St, d) \Rightarrow^* (Cr, (d_e \cdot St), d)$.

```

⇒*e : (e : E 1) → (d1 d2 Cr St : D)
    → eval e (updateE zero d1 initialVec) ≡ d2
    → < codeE e • Cr , St , d1 > ⇒* < Cr , d2 • St , d1 >

```


After that, we can prove that if for $C \in \mathbb{C}$, $C \vdash [V1 \mapsto d_1] \Rightarrow [V1 \mapsto d_2]$, then $((codeC\ C \cdot Cr), St, d_1) \Rightarrow^* (Cr, St, d_2)$.

```

=>*ok : (c : C 1) → (d1 d2 Cr St : D) → (out : Vec D 1)
  → c ⊢ updateE zero d1 initialVec ⇒ out
  → dlookup zero out ≡ d2
  → < codeC c • Cr , St , d1 > ⇒* < Cr , St , d2 >

```

This proof shows that the execution of *command* has relationship with the relation \Rightarrow , which means that one step of execution of *command* is corresponding to the several step \Rightarrow^* accordingly, which is the simulation of *WHILE* program in *Agda*.

2.2.2.2 Interpretation in the universal WHILE program

Then we should define the real universal *WHILE* program. The most important part is the *STEP* Macro. Before we defining the Macro, we should define some constants in \mathbb{E} which could be found in B.6.

Then we can define *STEP* as a *command* in *Agda* :

```

STEP-I : C 8
STEP-I = the interpretation command
         following the definition of STEP Macro
         and the syntax of WHILE program

```

Then we can prove that the simulation held by *Agda* has one step correspondence with the Macro *STEP*. That is, if $(Cd, St, V1) \Rightarrow (Cd', St', V1')$, then $STEP \vdash [\dots, Cd, St, V1, \dots] \Rightarrow [\dots, Cd', St', V1', \dots]$.

```

c-h : {Pd P C : D}(d1 d2 Cr1 Cr2 St1 St2 : D)
  → < Cr1 , St1 , d1 > ⇒ < Cr2 , St2 , d2 >
  → STEP-I ⊢ (Pd :: P :: C :: Cr1 :: St1 :: d1 :: dnil :: dnil :: [])
    ⇒ (Pd :: P :: C :: Cr2 :: St2 :: d2 :: dnil :: dnil :: [])

```

Because both the relation \Rightarrow and the execution of *while* loop does the induction on one step, we can prove that the several steps relation in *Agda* simulation have correspondence with the Macro *STEP*.

```

step-I-ok : (c : C 1) → (d1 d2 : D)
  → < codeC c • dnil , dnil , d1 > ⇒* < dnil , dnil , d2 >
  → while (var Cd) STEP-I ⊢ (codeP (prog zero c zero)) • d1 ::
    (codeP (prog zero c zero)) ::

```

```

codeC {1} c :: codeC {1} c • dnil ::
dnil :: d1 :: dnil :: dnil :: []
⇒ (codeP (prog zero c zero)) • d1 ::
  (codeP (prog zero c zero)) ::
codeC {1} c :: dnil :: dnil ::
d2 :: dnil :: dnil :: []

```

From this proof we can know that if *Agda* have the ability of simulating some *WHILE* program, then the universal *WHILE* program can simulate the same *WHILE* program.

2.2.3 Correctness of the Universal WHILE model

Finally by using the proof that were mentioned above, we can prove the correctness of the universal *WHILE* program.

```

execP-uni : (p : P 1) → (d1 d2 : D)
           → ExecP P d1 d2
           → ExecP universalI ((codeP p) • d1) d2

```

As a result, we can conclude that for $p \in \mathbb{P}$ and $inp, out \in \mathbb{D}$, if $p(inp) \equiv output$, then the universal *WHILE* program u , $u([p] \bullet inp) \equiv output$

2.3 Proof to the Halting Problem

2.3.1 Construct the WHILE program U

To prove the undecidability of the *Halting Problem*, we should construct a special *WHILE* program at first. Following the definition on *wiki*[?] and on *paper*[], we could construct the program **U** defined as following. The strategy to construct **U** is mentioned in 1.3.1. However, by considering the *syntax* and *semantic* of *WHILE* program and the universal *WHILE* program, we know that we must feed the code of another program into our universal program as part of the argument. That is, when we are constructing a program **U**, and assume that there is a program h that could decide the *Halting Problem*, then the argument to the program **U** should always be $([h] \bullet input)$. And to unify the argument to the program, the program h inside **U** should run on $(input \bullet ([h] \bullet input))$. Thus, the definition of program **U** in *Agda* is:

```

U : P 8
U = prog PD ((Pp := hd (var PD))

```

```

→→
(Cc := hd (tl (var Pp)))
→→
(Cd := cons (var Cc) nil)
→→
(St := nil)
→→
(V1 := cons (tl (var PD)) (cons (var Pp) (tl (var PD))))
→→
(while (var Cd) STEP-I)
→→
if-else8 (var V1) (while (cons nil nil) (V1 := var V1))
              (V1 := var V1)
)
V1

```

Because we don't have empty *command*, we use the command $x := \text{var } x$ to assign the same value to its original variable, to indicate the empty *command*.

2.3.1.1 Property 1 of U

From our definition of the program **U**, we can prove that if the execution result of $h(\text{input} \bullet ([h] \bullet \text{input}))$ is *true*, then if we feed h to the program **U** and execute the program **U** on $([h] \bullet \text{input})$, the program **U** will never terminate.

Initially we can prove that the infinite loop can't terminate, and if there is *command* in the form of *while true command*, then this *while* loop is an infinite loop.

```

wt : {n : ℕ} → C n → C n
wt c = while (cons nil nil) c

```

```

wt-loop : {t : D}{n : ℕ}{c : C n}{env1 env2 : Vec D n}
          → (p : wt c ⊢ env1 ⇒ env2) → loop-ct p ≡ t → ⊥
wt-loop (whilef ()) x
wt-loop {dnil} (whilet x p p1) ()
wt-loop {.(loop-ct p) • .(loop-ct p1)} (whilet x p p1) refl
      = wt-loop {loop-ct p1} p1 refl

```

Here we use *nil* to indicate *false* in *WHILE* and *others* to indicate *true* in *WHILE* program. The proof function does induction on the *call tree*, which means that the *assignment* is the leaf of the tree, *sequence* and *while* loop both have two branches.

Then we can prove that for any $h \in \mathbb{P}$, if $h(input \bullet ([h] \bullet input))$ yields *true*, then the execution of \mathbf{U} on $([h] \bullet input)$ will never terminate.

$$\begin{aligned} \text{execP-U-loop} : \{h : \mathbb{P}\} &\rightarrow (d_1 \ d_2 : \mathbb{D}) \\ &\rightarrow \text{ExecP } h \ (d_1 \bullet ((\text{codeP } h) \bullet d_1)) \ d_2 \\ &\rightarrow (d_2 \equiv \text{dnil} \rightarrow \perp) \\ &\rightarrow (\forall \{d_3 : \mathbb{D}\} \rightarrow \text{ExecP } U \ ((\text{codeP } h) \bullet d_1) \ d_3 \rightarrow \perp) \end{aligned}$$

2.3.1.2 Property 2 of U

From our definition of the program \mathbf{U} , we can prove that if the execution result of $h(input \bullet ([h] \bullet input))$ is *false*, then if we feed h to the program \mathbf{U} and execute the program \mathbf{U} on $([h] \bullet input)$, the program \mathbf{U} will terminate immediately.

$$\begin{aligned} \text{execP-U-halt} : \{h : \mathbb{P}\} &\rightarrow (d_1 \ d_2 : \mathbb{D}) \\ &\rightarrow \text{ExecP } h \ (d_1 \bullet ((\text{codeP } h) \bullet d_1)) \ d_2 \\ &\rightarrow d_2 \equiv \text{dnil} \\ &\rightarrow \text{ExecP } U \ ((\text{codeP } h) \bullet d_1) \ d_2 \end{aligned}$$

2.3.2 Proof the Undecidability of the Halting Problem

Finally we assume that there exists some program h that will decide the *Halting Problem* following the definition in 1.2.3.

2.3.2.1 Property 1 of Machine H

The program h is a program of *WHILE* which has the property that for all $p \in \mathbb{P}$ and $input \in \mathbb{D}$, if p **halts** on inp , then $h([p] \bullet input)$ yields *true*.

$$\begin{aligned} \text{prop}_1 : \forall \{n : \mathbb{N}\} &\rightarrow \forall \{p : \mathbb{P}\ n\} \rightarrow \forall \{inp : \mathbb{D}\} \\ &\rightarrow (\Sigma \ D \ (\text{ExecP } p \ inp) \\ &\rightarrow \text{ExecP } h \ ((\text{codeP } p) \bullet inp) \ \text{dtrue}) \end{aligned}$$

2.3.2.2 Property 2 of Machine H

The program h also has the property that for all $p \in \mathbb{P}$ and $input \in \mathbb{D}$, if p doesn't **halt** on inp , then $h([p] \bullet input)$ yields *false*.

$$\begin{aligned} \text{prop}_2 : \forall \{n : \mathbb{N}\} &\rightarrow \forall \{p : \mathbb{P}\ n\} \rightarrow \forall \{inp : \mathbb{D}\} \\ &\rightarrow (\forall \{out : \mathbb{D}\} \rightarrow \text{ExecP } p \ inp \ out \rightarrow \perp) \\ &\rightarrow \text{ExecP } h \ ((\text{codeP } p) \bullet inp) \ \text{dfalse} \end{aligned}$$

2.3.2.3 Propositional Proof

Then we can abstract the proof from the two properties of the program \mathbf{U} and the program h . We can name the property " \mathbf{U} halt on $(\lfloor h \rfloor \bullet \text{input})$ " as X , " $h(\lfloor p \rfloor \bullet \text{input})$ yielding *true*" as Y and " $h(\lfloor p \rfloor \bullet \text{input})$ yielding *false*" as Z . Then we can rename the two properties of \mathbf{U} as xy and nxz , and the two properties of h as ynx and zx . Note that $\neg \exists x, Px \equiv \forall x, \neg Px$. Then we can get *false* from those four propositions.

postulate

```
X Y Z : Set
xy  : X → Y
nxz : (X → ⊥) → Z
ynx : Y → X → ⊥
zx  : Z → X
```

```
a⊥ : X → ⊥
a⊥ a = ynx (xy a) a
```

```
bot : ⊥
bot = a⊥ (zx (nxz a⊥))
```

2.3.2.4 Final Proof

Finally we can prove the *undecidability* of the *Halting Problem*, which means we assume there exists a program h which could *decide* the *Halting Problem* and get \perp from our assumption.

```
halt-contradiction : {h : P 1}
  → (∀ {n : ℕ} → ∀ {p : P n} → ∀ {inp : D}
    → (Σ D (ExecP p inp)
      → ExecP h ((codeP p) • inp) dtrue)
    × ((∀ {out : D} → ExecP p inp out → ⊥)
      → ExecP h ((codeP p) • inp) dfalse))
  → ⊥
halt-contradiction {h} p = exec-U-⊥ (dnil ,
  (execP-U-halt {h} ((codeP U)) dnil
    (u-loop
      (lambda {out} q
        → exec-U-⊥ (out , q)))
    refl))
```

where

$\text{prop} = p \{8\}\{U\}\{((\text{codeP } h) \bullet (\text{codeP } U))\}$

$\text{u-halt} : \Sigma D (\text{ExecP } U ((\text{codeP } h) \bullet (\text{codeP } U)))$
 $\rightarrow \text{ExecP } h ((\text{codeP } U) \bullet ((\text{codeP } h) \bullet (\text{codeP } U))) \text{ dtrue}$
 $\text{u-halt} = \text{proj}_1 \text{ prop}$

$\text{u-loop} : (\forall \{out : D\} \rightarrow \text{ExecP } U ((\text{codeP } h) \bullet (\text{codeP } U)) \text{ out} \rightarrow \perp)$
 $\rightarrow \text{ExecP } h ((\text{codeP } U) \bullet ((\text{codeP } h) \bullet (\text{codeP } U))) \text{ dfalse}$
 $\text{u-loop} = \text{proj}_2 \text{ prop}$

$\text{exec-U-}\perp : \Sigma D (\text{ExecP } U ((\text{codeP } h) \bullet (\text{codeP } U))) \rightarrow \perp$
 $\text{exec-U-}\perp (d, p) = \text{execP-U-loop } ((\text{codeP } U)) \text{ dtrue}$
 $(\text{u-halt } (d, p)) (\text{lambda } \{ () \} p$

Chapter 3

Evaluation, Summary and Further Work

3.1 Evaluation and Summary

This dissertation implements a formalisation of the *Turing-complete* computational model *WHILE* program, and the proof of the undecidability of the *Halting Problem* in *Agda*. It also helps readers to understand the *Church-Turing Thesis*, the *Turing Machine*, the universal computational model. Above all, it presents an accessible proof of the undecidability of the famous problem – the *Halting Problem*, and explains the details on how we can assume a hypothetical model and derive *false* from our assumption.

The idea of the *Turing-complete* model – *WHILE* language follows *Neil D. Jones's* work in the paper *Computability and Complexity: from a Programming Perspective*?, including a lot of personal effort and intelligence on the definition of the *WHILE* model in *Agda* and the construction of the universal *WHILE* program in *Agda*. The idea of the correctness of the universal *WHILE* model is obvious because the *WHILE* language satisfies the *Turing-complete* property. However, to prove the correctness of the universal *WHILE* model is extremely hard. The core part of the project is to find the correspondence between the execution of the *WHILE* program on some input and the universal *WHILE* program which takes the previous program as part of the argument. With the help of Dr Thorsten Altenkirch, the supervisor of this project, I developed an inspired idea which constructed the intermediate *Agda* interpretation step. Based on the intermediate *Agda* interpretation step, I constructed the relationship between the execution of the *WHILE* program and the universal *WHILE*

program. Additionally, the idea of the proof of the undecidability of the *Halting Problem* follows some of the *Alan Turing's* work[?]. I also develop my own version of the proving process and the details can be found in 2.3. Since our constructed *WHILE* program *U* must take the code of some other programs as the input, the argument to the hypothetical *WHILE* program *h* (the program presumable can decide the *Halting Problem*) can't just follow the proof on paper[?]. Furthermore, I also abstract the proof on propositional level which can be found in 2.3.2.3.

There are two aspects that using *Agda* to do this project. One is to develop a machine checked proof of the undecidability of the *Halting Problem*. The *Halting Problem* is such a famous problem that almost all the people who worked around with computer science are familiar with. However, people can only find the idea of the proof of the undecidability of the *Halting Problem* theoretically. There is no strict proof using *Agda*, the interactive proof assistant language, which should formalise a computational model is *Turing-complete* and use that model to derive the problem. That is, proving the undecidability of the *Halting Problem* on paper theoretically is easy by following the idea of the *Alan Turing's* work[?], but deriving the machine checked proof of the same problem is considerably difficult. The understanding of the method to construct the *Turing-complete* computational model and prove the correctness of that model, and the construction of the machine checked proof of the undecidability of the *Halting Problem* will help people know the details of the *Church-Turing Thesis*, the computational model and the *Halting Problem* distinctly instead of wandering at an abstract and theoretical level. The second aspects is that using *Agda* as the interactive proof assistant language has a lot of advantages. Since *Agda* allows us to use Unicode characters while we write programs in *Agda* and that may lead to the proofs in *Agda* look similar to those logic proofs on paper and textbooks. Thus, readers can go through the whole proof plainly. Combining with the previous aspect, people can read each line of the proof in depth as well as understand the idea of the construction of the machine checked proof, the general as well as the detail idea of the proof. Besides, *Agda* can automatically infer the type of the implicit argument (although sometimes which will cause the compilation of the program a little bit slow), pattern match on the inductive argument conveniently and show the type of each variable to the current goal. Furthermore, *Agda* can compute the type of the goal based on the definition of the function and current procedure of the program, check whether the provided 'answer' matches the type to the goal or not, and update the type to the consequence goals if some goals have been completed. Furthermore, lots of convenient built-in functions and the standard *Agda* library help me a lot, and make the whole

machine checked proof to the problem much easier.

The management to the project follows the plan proposed at the beginning of this academic year. I used *GitHub* as the program revision control and source code management system to manage the project. I determined the object to the project at the end of last summer vacation after I went through *Neil D. Jones's* paper and the *Agda* tutorial[?][?]. I started the project earlier than almost all the other people, and finished the construction of the *WHILE* language and the universal *WHILE* model in *Agda* before Christmas. Then I struggled with the proof of the correctness of the universal *WHILE* model for two months. At the beginning of this semester, I worked out a possible solution to the previous problem, and rushed the project along with the final proof of the undecidability of the *Halting Problem* in one month under Ambrus's help. I finished the code to the project at the beginning of March, and uploaded the final work to *GitHub*. Then I spent three weeks working out the final report of the project. Along my way, I insulted and made formal visits to my supervisor Altenkirch Thorsten once a week formally, and always talked to Ambrus Kaposi and Gabe Dijkstra to work out solution to the project all the time. Without their help, I can not finish the project on time. I learnt a lot from the procedure of the project, including the knowledge and technique to *Agda*, the knowledge in the area of the computability. Furthermore, I also figure down the way of managing a project, and manage my time to the project. During the process, my communication and writing skills have been improved a lot. Above all, this project means a lot to me and becomes one of the most significant parts of my undergraduate studys.

3.2 Main Difficulties

Since the main object – the *Halting Problem* to this project is a famous theoretical topic that is relevant to the computational theorem in computer science, and in our knowledge this project is the project that first time successfully formalised a *Turing-complete* computational model, and proved the undecidability of the *Halting Problem* in *Agda*, this project contains many difficulties and uncertainties.

Initially, choosing a proper computational model will determine the difficulty of the whole project. Constructing a *Turing Machine* in *Agda* is still hard even with the help of *Agda* as the interactive proof assistant language. I had also considered the *lambda-calculus* as the computational model in the proof. Finally I choose the *WHILE* language and the result shows that this was a wise decision.

Then proving the correctness of the universal *WHILE* program is the core of the

project, and it is thoroughly hard. As I said in 3.1 I found the intermediate step between the *WHILE* program and the universal *WHILE* model. In addition, even with the intermediate step, finding the relationships between *WHILE* and the intermediate step, and the relationships between the intermediate step and the universal *WHILE* model are still difficult. The execution of the *command* is a partial function and is implemented as a relationship in *Agda*. The interpretation of the *WHILE* program in *Agda* will violate the termination checker because of the possible situation of the infinite *while* loop. That is the reason I do induction on the *call tree* in the proof. Additionally, the one step relation between the intermediate step and the universal *WHILE* model is the most difficult part in the project. Initially I used a lot of implicit arguments, which caused *Agda* crashing on the calculation of the type inference. I used about thirty separate files, each file contains a helper function, including almost 4000 lines of code to finish that proof, which troubled me for two months. The final proof of the correctness of the universal *WHILE* model was too hard to prove at once, so I separated it into several steps which only makes it little bit easier.

Finally, the proof of the undecidability of the *Halting Problem* is really difficult. Since *Agda* will automatically compute the type of the goal, and expand the definition it currently used, it will cost hours of time to check the type of the goal. Without the abstraction to the final proof, I can hardly know the target of the final proof. For those reasons, the final proof is derived without auto goal type checking and auto goal type inference, and come down to a lot of basic logical rules, which makes it the one of the most difficult part in this machine checked proof.


3.3 Further Work

3.3.1 WHILE Program Variables Transformation

The universal *WHILE* program defined in 2.2 can only simulate the *WHILE* program that has only one variable. Thus the final proof can only be used in a condition when machine *h* has only one variable. Under certain circumstances, it then leads to the decision that the *Halting Problem* is *undecidable*. Additionally, the *WHILE* program that has only one variable (we name it as **WHILE-I** program) has the same computation ability compared to the *WHILE* program that has many variables. That means, the number of variables to the *WHILE* program does not matter actually, and doesn't violate the property of *Turing-completeness* to the *WHILE* program. However, since *Agda* has strict type, the program must know the number of its variables

before it was defined. Thus, we can construct some rules to transform the *WHILE* program to the **WHILE-I** program and prove that they have the same effect, which means that for $p \in \mathbb{P} \ n$ and $input, output \in \mathbb{D}$, $p(input)$ yielding $output$ implies that $\exists p-I \in \mathbb{P} \ 1$, $p-I(input)$ yielding $output$.

Initially we should transform the *environment* of the program from $\text{Vec } \mathbb{D} \ n$ to $\text{Vec } \mathbb{D} \ 1$, which means accumulating all the variables in the first *environment* to the first element in the second *environment* by the operator \cdot . For example, initially we have an *environment* of five variables: $(A :: B :: C :: D :: E :: [])$, then the transformation will construct a tree structure data for the first element of the new *environment*:




Then we can transform the *expression*. For the *expression* of *head*, *tail*, *cons*, *nil* and *equality*, we can easily recursively transform the target *expression* from the argument. For the *expression* that use the value of variable that in the *environment*, for example *var C* in our previous example, we can use the *expression* $\text{head} (\text{tail} (\text{tail} (\text{var zero})))$ to get the same value as the transformed 1 variable environment.

Then similarly we can transform the *command*. For the *command* of *sequence* and *while* loop, we can easily recursively transform the target *command* from the argument. For the *expression* of the *assignment*, for example if we assign variable *C* with value $K \in \mathbb{E}$, the *environment* with five variables would update as:

$$(A :: B :: C :: D :: E :: []) \Longrightarrow (A :: B :: \mathbf{K} :: D :: E :: [])$$

we can write the new *assignment command* as

$\text{zero} := (\text{hd} (\text{var zero})) \cdot ((\text{hd} (\text{tl} (\text{var zero}))) \cdot ((K) \cdot (\text{tl} (\text{tl} (\text{tl} (\text{var zero}))))))$ and



Finally we can transform the *program*. We should transform the initial *environment* at first. Then we can transform the *command*. Finally we should get the result from the transformed *environment*.

The proof of the correctness of the transformation will be achieved in the future.

3.3.2 Interpretation of the WHILE Program with Arbitrary Variables

If we can prove that the *WHILE* program has the same computation ability with **WHILE-I** program which has only one variable, we can conclude that our universal *WHILE* program defined in 2.2 can simulate the *WHILE* program with arbitrary variables. For example, for $p \in \mathbb{P} n$ and $input, output \in \mathbb{D}$, if we want to use our universal *WHILE* program to simulate $p(input)$, then we should transform p to $p-I$ which has only one variable by our predefined transformation function. And we know that $p(input)$ yielding $output$ implies that $p-I(input)$ yielding $output$ by the proof of correctness of the transformation function. And by the proof of the correctness of the universal *WHILE* program we know that $p-I(input)$ yielding $output$ implies that $u([p-I] \cdot input)$ yielding $output$. Thus we can conclude that for $p \in \mathbb{P} n$ and $input, output \in \mathbb{D}$, $p(input)$ yielding $output$ implies $u([p-I] \cdot input)$ yielding $output$ which means universal *WHILE* program defined in 2.2 can simulate the *WHILE* program has arbitrary variables.

Thus for the proof of *Halting Problem*, we can say that for $p \in \mathbb{P} n$, h can decide the *Halting Problem* implies *false*.

Appendix A

Definition of the WHILE Language

The definition of *WHILE* language used in the project follows the definition in the paper *Computability and complexity: from a programming perspective* by Neil D. Jones in 1997[?].

A.1 Syntax of the Expression

Expressions \ni E, F ::= X (for $X \in \text{Vars}$)
| d (for atom d , one atom *nil* defined in *Agda*)
| **cons** E F
| **hd** E
| **tl** E
| **=?** E F

A.2 Syntax of the Command

Commands \ni C, D ::= X := E
| C ; D
| **while** E **do** C

A.3 Semantics of the Expression

The definition of evaluation function ε is: for $e \in \mathbb{E}$ and a given *store* of program \mathbb{P} , $\sigma \in \text{Store}^{\mathbb{P}}$, $\varepsilon[e]\sigma = d \in \mathbb{D}$.

$$\begin{aligned}
\varepsilon[X]\sigma &= \sigma(X) \\
\varepsilon[d]\sigma &= d \\
\varepsilon[\text{cons } E \ F]\sigma &= \varepsilon[E]\sigma \cdot \varepsilon[F]\sigma \\
\varepsilon[\text{hd } E]\sigma &= \begin{cases} e & \text{if } \varepsilon[E]\sigma = (e, f) \\ \text{nil} & \text{otherwise} \end{cases} \\
\varepsilon[\text{tl } E]\sigma &= \begin{cases} f & \text{if } \varepsilon[E]\sigma = (e, f) \\ \text{nil} & \text{otherwise} \end{cases} \\
\varepsilon[=? \ E \ F]\sigma &= \begin{cases} \text{true} & \text{if } \varepsilon[E]\sigma = \varepsilon[F]\sigma \\ \text{false} & \text{otherwise} \end{cases}
\end{aligned}$$

A.4 Semantics of the Command

The definition of the execution relationship is: for $c \in \mathbb{C}$, $c \vdash \sigma \rightarrow \sigma' \subseteq \mathbb{C} \times \text{Store}^{\mathbb{P}} \times \text{Store}^{\mathbb{P}}$ where σ' is the new *environment* updated by the execution of *command* c .

$$\begin{aligned}
X := E \vdash \sigma \rightarrow \sigma[X \mapsto d] & \quad \text{if } \varepsilon[E]\sigma = d \\
C; D \vdash \sigma \rightarrow \sigma'' & \quad \text{if } C \vdash \sigma \rightarrow \sigma' \text{ and } D \vdash \sigma' \rightarrow \sigma'' \\
\text{while } E \text{ do } C \vdash \sigma \rightarrow \sigma'' & \quad \text{if } C \varepsilon[E]\sigma \neq \text{nil}, C \vdash \sigma \rightarrow \sigma', \\
& \quad \text{while } E \text{ do } C \vdash \sigma' \rightarrow \sigma'' \\
\text{while } E \text{ do } C \vdash \sigma \rightarrow \sigma & \quad \text{if } C \varepsilon[E]\sigma = \text{nil}
\end{aligned}$$

A.5 Example of the WHILE program

Here we give an example code on *WHILE* language and use the defined part to execute the program.

A.5.1 Example of the WHILE program in the WHILE model

The *WHILE* program **concat** which could concatenate two list into one define as below:

```

read X; (* X is (d.e) *)
  A := hd X; (* A is d *)
  Y := tl X; (* Y is e *)
  B := nil; (* B becomes d reversed *)
while A do

```

```

    B := cons (hd A) B;
    A := tl A;
  while B do
    Y := cons (hd B) Y;
    B := tl B;
  write Y

```

A.5.2 Example of the WHILE program in *Agda*

Here we construct the same program using the definition we defined in *Agda*, it should be in the following format:

```

append : P 4
append = prog zero
  ((suc (suc zero) := hd (var zero))
   →→
   (suc zero := tl (var zero))
   →→
   (suc (suc (suc zero)) := nil)
   →→
   (while
    (var (suc (suc zero)))
    ((suc (suc (suc zero)) :=
      cons (hd (var (suc (suc zero))))
          (var (suc (suc (suc zero)))))
     →→
     ((suc (suc zero)) := tl (var (suc (suc zero)))))
    →→
    (while
     (var (suc (suc (suc zero))))
     ((suc zero := cons (hd (var (suc (suc (suc zero)))))
                        (var (suc zero)))
      →→
      (suc (suc (suc zero)) := tl (var (suc (suc (suc zero)))))
     (suc zero)

```

A.5.3 Execution of the Example *WHILE* program in *Agda*

To execute the program, we define three lists (in format of \mathbb{D}) in which **list1** and **list2** are the two input lists and **list3** is the result:

```

list1 : D
list1 = ltod (1 :: 2 :: 3 :: [])

```

```
list2 : D
list2 = ltod (4 :: 5 :: 6 :: [])
```

```
list3 : D
list3 = ltod (1 :: 2 :: 3 :: 4 :: 5 :: 6 :: [])
```

Now we could execute the *WHILE* program using our definitions of syntax and semantics:

```
runAppend : ExecP append (list1 • list2) list3
runAppend = terminate zero (suc zero)
  {env = list1 • list2 :: list3 :: dnil :: dnil :: []}
(seq {env1 = list1 • list2 :: dnil :: dnil :: dnil :: []}
  assign
  (seq {env1 = list1 • list2 :: dnil :: list1 :: dnil :: []}
    assign
    (seq {env1 = list1 • list2 :: list2 :: list1 :: dnil :: []}
      assign
      (seq {env1 = list1 • list2 :: list2 :: list1 :: dnil :: []}
        {env2 = result}
        {env3 = list1 • list2 :: list3 :: dnil :: dnil :: []}
        (whilet {env1 = list1 • list2 :: list2 :: list1 :: dnil :: []}
          {env2 = list1 • list2 :: list2 ::
            dsnd list1 :: dfst list1 • dnil :: []}
          {env3 = result}
          tt
          (seq assign assign)
          (whilet {env1 = list1 • list2 :: list2 ::
            dsnd list1 :: dfst list1 • dnil :: []}
            {env2 = list1 • list2 :: list2 :: dsnd (dsnd list1) ::
              dfst (dsnd list1) • (dfst list1 • dnil) :: []}
            {env3 = result}
            tt
            (seq assign assign)
            (whilet {env1 = list1 • list2 :: list2 :: dsnd (dsnd list1) ::
              dfst (dsnd list1) • (dfst list1 • dnil) :: []}
              {env2 = result}
              {env3 = result}
              tt
              (seq assign assign)
              (whilet {env1 = result}
                {env2 = list1 • list2 :: const 3 • list2 ::
```



```

        dnil :: ltod (2 :: 1 :: []) :: []}
    {env3 = list1 • list2 :: list3 :: dnil :: dnil :: []}
  tt
  (seq assign assign)
  (whilet {env1 = list1 • list2 :: const 3 • list2 ::
    dnil :: ltod (2 :: 1 :: []) :: []}
    {env2 = list1 • list2 :: const 2 • (const 3 • list2) ::
    dnil :: ltod (1 :: []) :: []}
    {env3 = list1 • list2 :: list3 :: dnil :: dnil :: []}
  tt
  (seq assign assign)
  (whilet {env1 = list1 • list2 :: const 2 • (const 3 • list2) ::
    dnil :: ltod (1 :: []) :: []}
    {env2 = list1 • list2 :: list3 :: dnil :: dnil :: []}
    {env3 = list1 • list2 :: list3 :: dnil :: dnil :: []}
  tt
  (seq assign assign)
  (whilef tt)))))))))
where
  result : Vec D 4
  result = list1 • list2 :: list2 ::
    dnil :: ltod (3 :: 2 :: 1 :: []) :: []

```

Appendix B

Definition of the Universal WHILE model

The definition of the universal *WHILE* language used in the project follows the definition in the paper *Computability and complexity: from a programming perspective* by Neil D. Jones in 1997[?].

B.1 Constant in WHILE

First of all, we define a function that would construct some value in data of \mathbb{D} based on the natural number \mathbb{N} :

```
const : (n :  $\mathbb{N}$ )  $\rightarrow$  D
const zero = dnil
const (suc n) = (dnil • dnil) • const n
```

Then, there are sixteen constants in \mathbb{D} that are used to indicate special meaning in the universal *WHILE* program.

```
dquote : D
dquote = const 1
```

```
d:= : D
d:= = const 2
```

```
d $\rightarrow\rightarrow$  : D
d $\rightarrow\rightarrow$  = const 3
```

```
dwhile : D
dwhile = const 4
```

```

dvar : D
dvar = const 5

ddnil : D
ddnil = const 6

dcons : D
dcons = const 7

dhd : D
dhd = const 8

dtl : D
dtl = const 9

d=? : D
d=? = const 10

dohd : D
dohd = const 11

dotl : D
dotl = const 12

docons : D
docons = const 13

doasgn : D
doasgn = const 14

dowh : D
dowh = const 15

do=? : D
do=? = const 16

```

B.2 Code the WHILE program

The mapping function from the *WHILE* program to *WHILE* data is defined as:

$$\underline{\text{read } V_i; C; \text{write } V_j} = ((\text{vari})\underline{C}(\text{varj}))$$

$$\underline{C; D} = (;\underline{CD})$$

$$\begin{array}{ll}
\underline{\text{while E do C}} & = \quad (\text{while}\underline{\text{EC}}) \\
\underline{\text{V}_i := \text{E}} & = \quad (:=(\text{vari})\underline{\text{E}}) \\
\\
\underline{\text{V}_i} & = \quad (\text{vari}) \\
\underline{\text{d}} & = \quad (\text{quoted}) \\
\underline{\text{cons E F}} & = \quad (\text{cons}\underline{\text{EF}}) \\
\underline{\text{hd E}} & = \quad (\text{hd}\underline{\text{E}}) \\
\underline{\text{tl E}} & = \quad (\text{tl}\underline{\text{E}}) \\
\underline{=? \text{ E F}} & = \quad (=?\underline{\text{EF}})
\end{array}$$

B.3 Variable of the Universal WHILE Program

We can define the eight variables of the universal *WHILE* program as:

PD : Fin 8

PD = zero

Pp : Fin 8

$$P_p = \text{suc zero}$$

Cc : Fin 8

$$C_c = \text{suc} (\text{suc} (\text{zero}))$$

Cd : Fin 8

$$\text{Cd} = \text{suc} (\text{suc} (\text{suc} (\text{zero})))$$

St : Fin 8

$$\text{St} = \text{suc} (\text{suc} (\text{suc} (\text{suc} (\text{zero}))))$$

V1 : Fin 8

```
V1 = suc (suc (suc (suc (suc (zero)))))
```

W : Fin 8

```
W = suc (suc (suc (suc (suc (suc (zero))))))
```

Z : Fin 8

$$Z = \text{suc} (\text{suc} (\text{suc} (\text{suc} (\text{suc} (\text{suc} (\text{suc} (\text{zero})))))))$$

B.4 Universal WHILE Program

The universal *WHILE* program which could simulate the *WHILE* program of 1 variable is defined as:

```

read PD;                (* Input (p.d) *)
  P := hd PD;            (* P = ((var 1) C (var 1)) *)
  C := hd (tl P)         (* C = hd tl p program code is C *)
  Cd := cons C nil;      (* Cd = (c.nil), Code to execute is c *)
  St := nil;             (* St = nil, Stack empty *)
  V1 := tl PD;           (* V1 = d Initial value of var. *)
  while Cd do STEP;      (* do while there is code to execute *)
write V1;

```

Where the *STEP* is the Macro that simulates the program.

B.5 STEP Macro

When we are simulating a *WHILE* program which has only one variable, we can easily define a stack machine as a Macro based on three variables: *Cd* which is the *command*, *St* which is the stack and *V1* which is the only one variable. Here we define the syntax sugar *cons** as *cons* A B C = cons A (cons B C)*. Then we could rewrite $[Cd, St]$ by:

$[((\text{quote } D) \cdot Cr), St]$	\Rightarrow	$[Cr, \text{cons } D \text{ } St]$
$[((\text{var } 1) \cdot Cr), St]$	\Rightarrow	$[Cr, \text{cons } V1 \text{ } St]$
$[((\text{hd } E) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E \text{ dohd } Cr, Sr]$
$[(\text{dohd} \cdot Cr), (T \cdot Sr)]$	\Rightarrow	$[Cr, \text{cons } (\text{hd } T) \text{ } St]$
$[((\text{tl } E) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E \text{ dotl } Cr, St]$
$[(\text{dotl} \cdot Cr), (T \cdot Sr)]$	\Rightarrow	$[Cr, \text{cons } (\text{tl } T) \text{ } Sr]$
$[((\text{cons } E_1 \ E_2) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E_1 \ E_2 \text{ docons } Cr, St]$
$[(\text{docons} \cdot Cr), (U \cdot (T \cdot Sr))]$	\Rightarrow	$[Cr, \text{cons } (\text{cons } T \ U) \text{ } Sr]$
$[((=? \ E_1 \ E_2) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E_1 \ E_2 \text{ do=? } Cr, St]$
$[(\text{do=?} \cdot Cr), (U \cdot (T \cdot Sr))]$	\Rightarrow	$[Cr, \text{cons } (=? \ T \ U) \text{ } St]$
$[((; \ C_1 \ C_2) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* C_1 \ C_2 \text{ } Cr, St]$
$[((:= \ (\text{var } 1) \ E) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E \text{ doasgn } Cr, St]$
$[(\text{doasgn} \cdot Cr), (W \cdot Sr)]$	\Rightarrow	$\{Cd := Cr, St := Sr; V1 := W\}$
$[((\text{while } E \ C) \cdot Cr), St]$	\Rightarrow	$[\text{cons}^* E \text{ dowh } (\text{while } E \ C) \text{ } Cr, St]$

$$\begin{aligned}
[(\text{dowh} \cdot ((\text{while } E \ C) \cdot \text{Cr})), (\text{nil} \cdot \text{Sr})] &\Rightarrow [\text{Cr}, \text{Sr}] \\
[(\text{dowh} \cdot ((\text{while } E \ C) \cdot \text{Cr})), ((A \cdot B) \cdot \text{Sr})] &\Rightarrow [\text{cons}^* \ C \ (\text{while } E \ C) \ \text{Cr}, \text{Sr}] \\
[\text{nil}, \text{St}] &\Rightarrow [\text{nil}, \text{St}]
\end{aligned}$$

B.6 Constant in the Expression

First of all, we define a function that would construct some value in the format of \mathbb{E} based on the data of \mathbb{D} :

```

dtoE : {n : ℕ} → D → E n
dtoE dnil = nil
dtoE (d1 • d2) = cons (dtoE d1) (dtoE d2)

```

Then, there are sixteen constant in \mathbb{E} that are used to indicate special meaning in the universal *WHILE* program.

```

quoteE : {n : ℕ} → E n
quoteE = dtoE dquote

```

```

varE : {n : ℕ} → E n
varE = dtoE dvar

```

```

valueE : {n : ℕ} → (f : Fin n) → E n
valueE f = dtoE (dftod f)

```

```

hdE : {n : ℕ} → E n
hdE = dtoE dhd

```

```

dohdE : {n : ℕ} → E n
dohdE = dtoE dohd

```

```

tlE : {n : ℕ} → E n
tlE = dtoE dtl

```

```

dotlE : {n : ℕ} → E n
dotlE = dtoE dotl

```

```

consE : {n : ℕ} → E n
consE = dtoE dcons

```

```

doconsE : {n : ℕ} → E n
doconsE = dtoE docons

```

$$\begin{aligned} \text{=?E} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{=?E} &= \text{dtoE } d=? \end{aligned}$$

$$\begin{aligned} \text{do=?E} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{do=?E} &= \text{dtoE } \text{do=?} \end{aligned}$$

$$\begin{aligned} \rightarrow\rightarrow\text{E} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \rightarrow\rightarrow\text{E} &= \text{dtoE } d\rightarrow\rightarrow \end{aligned}$$

$$\begin{aligned} \text{:=E} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{:=E} &= \text{dtoE } d:= \end{aligned}$$

$$\begin{aligned} \text{doasgnE} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{doasgnE} &= \text{dtoE } \text{doasgn} \end{aligned}$$

$$\begin{aligned} \text{whileE} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{whileE} &= \text{dtoE } \text{dwhile} \end{aligned}$$

$$\begin{aligned} \text{dowhE} &: \{n : \mathbb{N}\} \rightarrow E \ n \\ \text{dowhE} &= \text{dtoE } \text{dowh} \end{aligned}$$